9

Use of Proficiency Levels

PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

INTRODUCTION

The values for student performance in reading, mathematics, and science literacy are usually considered as continuous latent variables. In order to facilitate the interpretation of the scores assigned to students, the reading, mathematics and science scales were designed to have an average score of 500 points and a standard deviation of 100 across OECD countries. This means that about two-thirds of the OECD member country students perform between 400 and 600 points.

In order to render PISA results more accessible to policy makers and educators, proficiency scales have been developed for the assessment domains. Since these scales are divided according to levels of difficulty and performance, both a ranking of student performance and a description of the skill associated with that proficiency level can be obtained. Each successive level is associated with tasks of increased difficulty.

In PISA 2000, five levels of reading proficiency were defined and reported in the PISA 2000 initial report *Knowledge and Skills for Life: First Results from PISA 2000* (OECD, 2001). In PISA 2003, six levels of mathematics proficiency levels were also defined and reported in the PISA 2003 initial report *Learning for Tomorrow's World – First Results from PISA 2003* (OECD, 2004a). In PISA 2006, six levels of science proficiency were defined and reported in the PISA 2006 initial report *Science Competencies for Tomorrow's World* (OECD, 2007a).

This chapter will show how to derive the proficiency levels from the PISA databases and how to use them.

GENERATION OF THE PROFICIENCY LEVELS

Proficiency levels are not included in the PISA databases, but they can be derived from the plausible values (PVs).

In PISA 2006, the cutpoints that frame the proficiency levels in science are 334.94, 409.54, 484.14, 558.73, 633.33 and 709.93.[1] While some researchers might understand that different possible scores can be assigned to a student, it is more difficult to understand that different levels can be assigned to a single student. Therefore, they might be tempted to compute the average of the five plausible values and then assign each student a proficiency level based on this average.

As discussed in Chapter 6 and Chapter 8, such a procedure is similar to assigning each student an expected *a posteriori* (EAP) score; the biases of such estimators are now well known. Since using EAP scores underestimates the standard deviation, the estimation of the percentages of students at each level of proficiency will consequently underestimate the percentages at the lowest and highest levels, and overestimate the percentages at the central levels.

As already stated, international education surveys do not aim to estimate the performance of particular students, but rather, they aim to describe population characteristics. Therefore, particular students can be allocated different proficiency levels for different plausible values. Thus, five plausible proficiency levels will be assigned to each student respectively according to their five plausible values. The SAS® syntax for the generation of the plausible proficiency levels in science is provided in Box 9.1.

The statement "array" allows the definition of a variable vector. In Box 9.1, two vectors are defined. The first, labelled SCIE, includes the five plausible values for the science combined scale (PV1SCIE to PV5SCIE) and the three science subscales (PV1EPS to PV5EPS, PV1ISI to PV5ISI, and PV1USE to PV5USE). The second, labelled LEVELSCIE, will create 20 new variables, labelled SCIELEV1 to SCIELEV5 for the science combined scale; EPSLEV1 to EPSLEV5 for the science/explaining phenomena scientifically subscale; ISILEV1 to ISILEV5 for the science/identifying scientific issues subscale; USELEV1 to USELEV1 to the science/use of scientific evidence subscale.

PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

Box 9.1 **SAS® syntax for generating the proficiency levels in science (*e.g.* PISA 2006)**

```
libname PISA2003 "c:\pisa\2003\data";
libname PISA2006 "c:\pisa\2006\data";
options nofmterr notes;
run;
data temp1;
        set pisa2006.stu;
        if (cnt="DEU");
        array scie (20)
           pv1scie pv2scie pv3scie pv4scie pv5scie
           pv1eps pv2eps pv3eps pv4eps pv5eps
           pv1isi pv2isi pv3isi pv4isi pv5isi
           pv1use pv2use pv3use pv4use pv5use;
        array levelscie (20)
           scielev1-scielev5
           epslev1-epslev5
           isilev1-isilev5
           uselev1-uselev5;
        do i=1 to 20;
           if (scie(i)<=334.94) then levelscie(i)=0;
           if (scie(i)>334.94 and scie(i)<=409.54) then levelscie(i)=1;
           if (scie(i)>409.54 and scie(i)<=484.14) then levelscie(i)=2;
           if (scie(i)>484.14 and scie(i)<=558.73) then levelscie(i)=3;
           if (scie(i)>558.73 and scie(i)<=633.33) then levelscie(i)=4;
           if (scie(i)>633.33 and scie(i)<=707.93) then levelscie(i)=5;
           if (scie(i)>707.93) then levelscie(i)=6;
        end;
        w_fstr0=w_fstuwt;
run;
```

The iterative process will recode each plausible value variable into a new variable with seven categories labelled 0 to 6 for science.

The computation of the percentage of students at each proficiency level and its respective standard error is similar to the computation of a mean estimate and its standard error, as described in Chapter 8, *i.e.*:

- For each plausible value, the percentage of students at each proficiency level and its respective standard error have to be computed. Per proficiency level, five percentage estimates denoted $\hat{\pi}_1$, $\hat{\pi}_2$, $\hat{\pi}_3$, $\hat{\pi}_4$ and $\hat{\pi}_5$ will be obtained. Out of the 80 replicates applied on each of the 5 proficiency level variables, per level of proficiency, 5 sampling variances will be estimated, denoted respectively $\sigma^2_{(\hat{\pi}_1)}$, $\sigma^2_{(\hat{\pi}_2)}$, $\sigma^2_{(\hat{\pi}_3)}$, $\sigma^2_{(\hat{\pi}_4)}$ and $\sigma^2_{(\hat{\pi}_5)}$. These five percentage estimates and their respective sampling variances are given in Table 9.1.
- The final percentage estimate is equal to the average of the five percentage estimates, *i.e.* $\hat{\pi} = \frac{1}{5}(\hat{\pi}_1 + \hat{\pi}_2 + \hat{\pi}_3 + \hat{\pi}_4 + \hat{\pi}_5)$
- The final sampling variance is equal to the average of the five sampling variances, *i.e.* $\sigma^2_{(\hat{\pi})} = \frac{1}{5}(\sigma^2_{(\hat{\pi}_1)} + \sigma^2_{(\hat{\pi}_2)} + \sigma^2_{(\hat{\pi}_3)} + \sigma^2_{(\hat{\pi}_4)} + \sigma^2_{(\hat{\pi}_5)})$
- The imputation variance, also denoted measurement error variance, is computed[2] as $\sigma^2_{(test)} = \frac{1}{4}\sum_{i=1}^{5}(\hat{\pi}_i - \hat{\pi})^2$
- The sampling variance and the imputation variance are combined to obtain the final error variance as $\sigma^2_{(error)} = \sigma^2_{(\hat{\pi})} + (1.2\,\sigma^2_{(test)})$
- The standard error is equal to the square root of the error variance.

This process is repeated for each proficiency level.

Table 9.1
The 405 percentage estimates for a particular proficiency level

Weight	PV1	PV2	PV3	PV4	PV5
Final	$\hat{\pi}_1$	$\hat{\pi}_2$	$\hat{\pi}_3$	$\hat{\pi}_4$	$\hat{\pi}_5$
Replicate 1	$\hat{\pi}_{1\_1}$	$\hat{\pi}_{2\_1}$	$\hat{\pi}_{3\_1}$	$\hat{\pi}_{4\_1}$	$\hat{\pi}_{5\_1}$
Replicate 2	$\hat{\pi}_{1\_2}$	$\hat{\pi}_{2\_2}$	$\hat{\pi}_{3\_2}$	$\hat{\pi}_{4\_2}$	$\hat{\pi}_{5\_2}$
Replicate 3	$\hat{\pi}_{1\_3}$	$\hat{\pi}_{2\_3}$	$\hat{\pi}_{3\_3}$	$\hat{\pi}_{4\_3}$	$\hat{\pi}_{5\_3}$
....	....	....	....	....	....
....	....	....	....	....	....
Replicate 80	$\hat{\pi}_{1\_80}$	$\hat{\pi}_{2\_80}$	$\hat{\pi}_{3\_80}$	$\hat{\pi}_{4\_80}$	$\hat{\pi}_{5\_80}$
Sampling variance	$\sigma^2_{(\hat{\pi}_1)}$	$\sigma^2_{(\hat{\pi}_2)}$	$\sigma^2_{(\hat{\pi}_3)}$	$\sigma^2_{(\hat{\pi}_4)}$	$\sigma^2_{(\hat{\pi}_5)}$

In this way 405 percentages will be estimated per proficiency level. As there are 7 levels in science, 2 835 percentages will be estimated.

The seven proficiency levels in science are:

1. Below level 1,
2. Level 1,
3. Level 2,
4. Level 3,
5. Level 4,
6. Level 5,
7. Level 6.

Sequentially applying the PROC_FREQ_NO_PV macro five times, as described in Chapter 7, will return, per proficiency level, five percentage estimates and five standard error estimates that can be combined to get the final estimate and its standard error.

Box 9.2 presents the SAS® syntax for sequentially running the PROC_FREQ_NO_PV macro five times. Table 9.2 presents the five estimates and their respective sampling variances, per proficiency level.

Box 9.2 **SAS® syntax for computing the percentages of students by proficiency level in science and its standard errors by using the PROC_FREQ_NO_PV macro (*e.g.* PISA 2006)**

```
%include "c:\pisa\macro\proc_freq_no_pv.sas";
%macro repeat;
%do kk=1 %to 5;
%BRR_FREQ( INFILE=temp1,
           REPLI_ROOT=w_fstr,
           BYVAR=cnt,
           VAR=scielev&kk,
           LIMIT=yes,
           LIMIT_CRITERIA=100 10 5 1,
           ID_SCHOOL=schoolid,
           OUTFILE=exercise&kk);
run;
%end;
%mend;
%repeat;
run;
```

To combine the results:

- Per proficiency level, the five percentage estimates are averaged.
- Per proficiency level, the five sampling variances are averaged.
- By comparing the final estimate and the five PV estimates, the imputation variance is computed.
- The final sampling variance and the imputation variance are combined as usual to get the final error variance.
- The standard error is obtained by taking the square root of the error variance.

Table 9.2
Estimates and sampling variances per proficiency level in science for Germany (PISA 2006)

Level		PV1	PV2	PV3	PV4	PV5
Below Level 1	$\hat{\pi}_i$	4.12	3.82	4.25	4.03	4.13
	$\sigma^2_{(\hat{\pi}_i)}$	$(0.60)^2$	$(0.59)^2$	$(0.72)^2$	$(0.67)^2$	$(0.71)^2$
Level 1	$\hat{\pi}_i$	11.93	11.8	11.03	11.09	10.70
	$\sigma^2_{(\hat{\pi}_i)}$	$(0.86)^2$	$(0.81)^2$	$(0.72)^2$	$(0.73)^2$	$(0.71)^2$
Level 2	$\hat{\pi}_i$	20.26	21.59	21.87	21.16	21.91
	$\sigma^2_{(\hat{\pi}_i)}$	$(0.73)^2$	$(0.71)^2$	$(0.73)^2$	$(0.80)^2$	$(0.76)^2$
Level 3	$\hat{\pi}_i$	28.70	27.46	27.33	27.92	27.93
	$\sigma^2_{(\hat{\pi}_i)}$	$(1.00)^2$	$(0.94)^2$	$(0.69)^2$	$(0.91)^2$	$(0.92)^2$
Level 4	$\hat{\pi}_i$	23.39	23.45	23.57	23.66	23.77
	$\sigma^2_{(\hat{\pi}_i)}$	$(0.91)^2$	$(0.93)^2$	$(0.92)^2$	$(0.92)^2$	$(0.96)^2$
Level 5	$\hat{\pi}_i$	9.82	10.07	10.14	10.24	9.69
	$\sigma^2_{(\hat{\pi}_i)}$	$(0.61)^2$	$(0.53)^2$	$(0.53)^2$	$(0.64)^2$	$(0.49)^2$
Level 6	$\hat{\pi}_i$	1.79	1.81	1.82	1.89	1.87
	$\sigma^2_{(\hat{\pi}_i)}$	$(0.25)^2$	$(0.28)^2$	$(0.23)^2$	$(0.20)^2$	$(0.21)^2$

The final results are presented in Table 9.3.

Table 9.3
Final estimates of the percentage of students, per proficiency level, in science and its standard errors for Germany (PISA 2006)

Proficiency level	%	S.E.
Below Level 1	4.07	0.68
Level 1	11.31	0.96
Level 2	21.36	1.06
Level 3	27.87	1.07
Level 4	23.57	0.95
Level 5	9.99	0.62
Level 6	1.84	0.24

A SAS® macro has been developed for computing the percentage of students at each proficiency level as well as its respective standard error in one run. Box 9.3 presents the SAS® syntax for running the macro and Table 9.4 presents the structure of the output data file.

Box 9.3 **SAS® syntax for computing the percentage of students by proficiency level in science and its standard errors by using the PROC_FREQ_PV macro (*e.g.* PISA 2006)**

```
%include "c:\pisa\macro\proc_freq_pv.sas";

%BRR_FREQ_PV(  INFILE=temp1,
               REPLI_ROOT=w_fstr,
               BYVAR=cnt,
               PV_ROOT=scielev,
               LIMIT=yes,
               LIMIT_CRITERIA=100 10 5 1,
               ID_SCHOOL=schoolid,
               OUTFILE=exercise6);
run;
```

This macro has eight arguments. Besides the usual arguments, the root of the proficiency level variable names has to be specified. For the science scale, as specified in the data statement of Box 9.1, this will be set as SCIELEV. As indicated in Table 9.4, the number of cases at Level 6 is less than 100.

Table 9.4
Output data file exercise6 from Box 9.3

CNT	SCIELEV	STAT	SESTAT	STUD_FLAG	SCH_FLAG	PCT_FLAG
DEU	0	4.07	0.68	0	0	1
DEU	1	11.31	0.96	0	0	0
DEU	2	21.36	1.06	0	0	0
DEU	3	27.87	1.07	0	0	0
DEU	4	23.57	0.95	0	0	0
DEU	5	9.99	0.62	0	0	0
DEU	6	1.84	0.24	1	0	1

As before, several breakdown variables can be used. For instance, the distribution of students across proficiency levels per gender can be obtained, as in Box 9.4.

Box 9.4 **SAS® syntax for computing the percentage of students by proficiency level and its standard errors by gender (*e.g.* PISA 2006)**

```
%BRR_FREQ_PV(INFILE=temp1,
             REPLI_ROOT=w_fstr,
             BYVAR=cnt st04q01,
             PV_ROOT=scielev,
             LIMIT=yes,
             LIMIT_CRITERIA=100 10 5 1,
             ID_SCHOOL=schoolid,
             OUTFILE=exercise7);
run;
```

In this case, the sum of the percentages will be equal to 100 per country and per gender, as shown in Table 9.5.

Table 9.5
Output data file exercise7 from Box 9.4

CNT	ST04Q01	SCIELEV	STAT	SESTAT	STUD_FLAG	SCH_FLAG	PCT_FLAG
DEU	1	0	3.73	0.67	1	0	1
DEU	1	1	12.12	1.19	0	0	0
DEU	1	2	21.08	1.26	0	0	0
DEU	1	3	29.94	1.47	0	0	0
DEU	1	4	23.29	1.07	0	0	0
DEU	1	5	8.42	0.73	0	0	1
DEU	1	6	1.42	0.38	1	0	1
DEU	2	0	4.39	0.84	0	0	1
DEU	2	1	10.55	1.09	0	0	0
DEU	2	2	21.62	1.23	0	0	0
DEU	2	3	25.93	1.21	0	0	0
DEU	2	4	23.83	1.35	0	0	0
DEU	2	5	11.46	1.03	0	0	0
DEU	2	6	2.22	0.37	1	0	1

As shown in Table 9.5, the percentage of males at Level 5 and 6 is higher than the percentage of females at Level 5 and 6.

The statistical significance of these differences cannot be evaluated with this procedure, however. More details on this issue will be provided in Chapter 11.

OTHER ANALYSES WITH PROFICIENCY LEVELS

Proficiency levels constitute a powerful tool for communicating the results on the cognitive test. Researchers and/or policy makers might therefore be interested in estimating the influence of some variables (such as the social background or self-confidence measures) on the proficiency levels.

PISA 2003, for instance, constructed an index of mathematics self-efficacy, denoted MATHEFF.

Analysing the relationship between proficiency levels and mathematics self-efficacy is relevant, as there is probably a reciprocal relationship between these two concepts. Better self-perception in mathematics is thought to increase a student's proficiency in mathematics, but an increase in the latter might in return affect the former.

Suppose that the statistic of interest is the average self-efficacy per proficiency level. In statistical terms, mathematics self-efficacy is considered as the dependent variable and the level of proficiency, the independent variable. There is no macro that can directly compute the mean of a continuous variable per proficiency level. On the other hand, the PROC_MEAN_NO_PV macro described in Chapter 7 can be applied sequentially five times and the results could be combined in an Excel® spreadsheet for instance. This will be the case whenever proficiency levels are used as independent or as classification variables.

Box 9.5 presents SAS® syntax for preparing the PISA 2003 data file.

Box 9.5 **SAS® syntax for generating the proficiency levels in mathematics (*e.g.* PISA 2003)**

```
data temp2;
        set pisa2003.stud;
        if (cnt=“DEU”);
        array math (25)
           pv1math pv2math pv3math pv4math pv5math
           pv1math1 pv2math1 pv3math1 pv4math1 pv5math1
           pv1math2 pv2math2 pv3math2 pv4math2 pv5math2
           pv1math3 pv2math3 pv3math3 pv4math3 pv5math3
           pv1math4 pv2math4 pv3math4 pv4math4 pv5math4;
        array levelmat (25)
           mlev1-mlev5
           m1lev1-m1lev5
           m2lev1-m2lev5
           m3lev1-m3lev5
           m4lev1-m4lev5;
        do i=1 to 25;
           if (math(i)<=357.77) then levelmat(i)=0;
           if (math(i)>357.77 and math(i)<=420.07) then levelmat(i)=1;
           if (math(i)>420.07 and math(i)<=482.38) then levelmat(i)=2;
           if (math(i)>482.38 and math(i)<=544.68) then levelmat(i)=3;
           if (math(i)>544.68 and math(i)<=606.99) then levelmat(i)=4;
           if (math(i)>606.99 and math(i)<=669.30) then levelmat(i)=5;
           if (math(i)>669.30) then levelmat(i)=6;
        end;
        w_fstr0=w_fstuwt;
        keep   cnt schoolid stidstd
               w_fstr0-w_fstr80
               mlev1-mlev5
               m1lev1-m1lev5
               m2lev1-m2lev5
               m3lev1-m3lev5
               m4lev1-m4lev5
               st03q01 matheff;
run;
```


PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

Box 9.6 presents SAS® syntax for computing the mean of student self-efficacy per proficiency level.

Box 9.6 **SAS® syntax for computing the mean of self-efficacy in mathematics and its standard errors by proficiency level (*e.g.* PISA 2003)**

```
%include "c:\pisa\macro\proc_means_no_pv.sas";
%macro repeat;
%do kk=1 %to 5;
%BRR_PROCMEAN(INFILE=temp2,
              REPLI_ROOT=w_fstr,
              BYVAR=cnt mlev&kk,
              VAR=matheff,
              STAT=mean,
              LIMIT=no,
              LIMIT_CRITERIA=,
              ID_SCHOOL=,
              OUTFILE=exercise&kk);
run;
data exercise&kk;
        set exercise&kk;
        stat&kk=stat;
        sestat&kk=sestat;
        mlev=mlev&kk;
        keep cnt mlev stat&kk sestat&kk;
run;
%end;
data exercise8;
        merge exercise1 exercise2 exercise3 exercise4 exercise5;
        by cnt mlev;
        stat=(stat1+stat2+stat3+stat4+stat5)/5;
        samp=  ((sestat1**2)+(sestat2**2)+(sestat3**2)+(sestat4**2)+
               (sestat5**2))/5;

        mesvar=(((stat1-stat)**2)+((stat2-stat)**2)+((stat3-stat)**2)+
                ((stat4-stat)**2)+((stat5-stat)**2))/4;

        sestat=(samp+(1.2*mesvar))**0.5;
        keep cnt mlev stat sestet stud_flag sch_flag pct_flag;
run;
%mend repeat;
```

Table 9.6 presents the mean estimates and standard errors for self-efficacy in mathematics per proficiency level.

To combine the results:

- Per proficiency level, the five mean estimates are averaged.
- Per proficiency level, the five sampling variances are averaged.
- By comparing the final estimate and the five PV estimates, the imputation variance is computed.
- The final sampling variance and the imputation variance are combined as usual to get the final error variance.
- The standard error is obtained by taking the square root of the error variance.

Final results are presented in Table 9.7. It shows that high self-efficacy in mathematics (STAT) is associated with higher proficiency levels (MLEV).

Table 9.6
Mean estimates and standard errors for self-efficacy in mathematics per proficiency level (PISA 2003)

Level		PV1	PV2	PV3	PV4	PV5
Below Level 1	$\hat{\mu}_i$	-0.68	-0.70	-0.74	-0.72	-0.77
	$\sigma^2_{(\hat{\mu}_i)}$	$(0.06)^2$	$(0.06)^2$	$(0.06)^2$	$(0.05)^2$	$(0.06)^2$
Level 1	$\hat{\mu}_i$	-0.44	-0.45	-0.42	-0.43	-0.40
	$\sigma^2_{(\hat{\mu}_i)}$	$(0.06)^2$	$(0.05)^2$	$(0.06)^2$	$(0.04)^2$	$(0.05)^2$
Level 2	$\hat{\mu}_i$	-0.18	-0.16	-0.17	-0.18	-0.18
	$\sigma^2_{(\hat{\mu}_i)}$	$(0.03)^2$	$(0.03)^2$	$(0.03)^2$	$(0.03)^2$	$(0.03)^2$
Level 3	$\hat{\mu}_i$	0.09	0.09	0.12	0.11	0.10
	$\sigma^2_{(\hat{\mu}_i)}$	$(0.03)^2$	$(0.03)^2$	$(0.03)^2$	$(0.03)^2$	$(0.03)^2$
Level 4	$\hat{\mu}_i$	0.43	0.45	0.41	0.45	0.44
	$\sigma^2_{(\hat{\mu}_i)}$	$(0.03)^2$	$(0.03)^2$	$(0.03)^2$	$(0.03)^2$	$(0.03)^2$
Level 5	$\hat{\mu}_i$	0.85	0.84	0.86	0.79	0.82
	$\sigma^2_{(\hat{\mu}_i)}$	$(0.04)^2$	$(0.04)^2$	$(0.03)^2$	$(0.04)^2$	$(0.04)^2$
Level 6	$\hat{\mu}_i$	1.22	1.23	1.27	1.28	1.29
	$\sigma^2_{(\hat{\mu}_i)}$	$(0.05)^2$	$(0.05)^2$	$(0.06)^2$	$(0.05)^2$	$(0.07)^2$

Table 9.7
Output data file exercise8 from Box 9.6

CNT	MLEV	STAT	SESTAT
DEU	0	-0.72	0.07
DEU	1	-0.43	0.06
DEU	2	-0.17	0.03
DEU	3	0.10	0.03
DEU	4	0.44	0.03
DEU	5	0.83	0.05
DEU	6	1.26	0.07

CONCLUSION

This chapter has shown how to compute the percentage of students per proficiency level and its standard errors. As shown, the algorithm is similar to the one used for other statistics.

The difficulty of conducting analyses using proficiency levels as the explanatory (independent) variables was also discussed.

PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

Notes

1. In PISA 2000, the cutpoints that frame the proficiency levels in reading are: 334.75, 407.47, 480.18, 552.89 and 625.61. In PISA 2003, the cutpoints that frame the proficiency levels in mathematics are: 357.77, 420.07, 482.38, 544.68, 606.99 and 669.3.

2. This formula is a simplification of the general formula provided in Chapter 5. M, denoting the number of plausible values, has been replaced by 5.

References

Beaton, A.E. (1987), *The NAEP 1983-1984 Technical Report*, Educational Testing Service, Princeton.

Beaton, A.E., *et al.* (1996), *Mathematics Achievement in the Middle School Years, IEA's Third International Mathematics and Science Study*, Boston College, Chestnut Hill, MA.

Bloom, B.S. (1979), *Caractéristiques individuelles et apprentissage scolaire*, Éditions Labor, Brussels.

Bressoux, P. (2008), *Modélisation statistique appliquée aux sciences sociales*, De Boek, Brussels.

Bryk, A.S. and **S.W. Raudenbush** (1992), *Hierarchical Linear Models for Social and Behavioural Research: Applications and Data Analysis Methods*, Sage Publications, Newbury Park, CA.

Buchmann, C. (2000), *Family structure, parental perceptions and child labor in Kenya: What factors determine who is enrolled in school?* a*Soc. Forces,* No. 78, pp. 1349-79.

Cochran, W.G. (1977), *Sampling Techniques*, J. Wiley and Sons, Inc., New York.

Dunn, O.J. (1961), "Multilple Comparisons among Menas", *Journal of the American Statistical Association*, Vol. 56, American Statistical Association, Alexandria, pp. 52-64.

Kish, L. (1995), *Survey Sampling*, J. Wiley and Sons, Inc., New York.

Knighton, T. and **P. Bussière** (2006), "Educational Outcomes at Age 19 Associated with Reading Ability at Age 15", Statistics Canada, Ottawa.

Gonzalez, E. and **A. Kennedy** (2003), *PIRLS 2001 User Guide for the International Database*, Boston College, Chestnut Hill, MA.

Ganzeboom, H.B.G., P.M. De Graaf and **D.J. Treiman** (1992), "A Standard International Socio-economic Index of Occupation Status", *Social Science Research* 21(1), Elsevier Ltd, pp 1-56.

Goldstein, H. (1995), *Multilevel Statistical Models,* 2nd Edition, Edward Arnold, London.

Goldstein, H. (1997), "Methods in School Effectiveness Research", *School Effectiveness and School Improvement* 8, Swets and Zeitlinger, Lisse, Netherlands, pp. 369-395.

Hubin, J.P. (ed.) (2007), *Les indicateurs de l'enseignement*, 2nd Edition, Ministère de la Communauté française, Brussels.

Husen, T. (1967), *International Study of Achievement in Mathematics: A Comparison of Twelve Countries,* Almqvist and Wiksells, Uppsala.

International Labour Organisation (ILO) (1990), *International Standard Classification of Occupations: ISCO-88.* Geneva: International Labour Office.

Lafontaine, D. and **C. Monseur** (forthcoming), "Impact of Test Characteristics on Gender Equity Indicators in the Assessment of Reading Comprehension", *European Educational Research Journal*, Special Issue on PISA and Gender.

Lietz, P. (2006), "A Meta-Analysis of Gender Differences in Reading Achievement at the Secondary Level", *Studies in Educational Evaluation* 32, pp. 317-344.

Monseur, C. and **M. Crahay** (forthcoming), "Composition académique et sociale des établissements, efficacité et inégalités scolaires : une comparaison internationale – Analyse secondaire des données PISA 2006", *Revue française de pédagogie*.

OECD (1998), *Education at a Glance – OECD Indicators,* OECD, Paris.

OECD (1999a), *Measuring Student Knowledge and Skills – A New Framework for Assessment,* OECD, Paris.

OECD (1999b), *Classifying Educational Programmes – Manual for ISCED-97 Implementation in OECD Countries,* OECD, Paris.

OECD (2001), *Knowledge and Skills for Life – First Results from PISA 2000,* OECD, Paris.

OECD (2002a), *Programme for International Student Assessment – Manual for the PISA 2000 Database,* OECD, Paris.

OECD (2002b), *Sample Tasks from the PISA 2000 Assessment – Reading, Mathematical and Scientific Literacy,* OECD, Paris.

OECD (2002c), *Programme for International Student Assessment – PISA 2000 Technical Report,* OECD, Paris.

OECD (2002d), *Reading for Change: Performance and Engagement across Countries – Results from PISA 2000,* OECD, Paris.

OECD (2003a), *Literacy Skills for the World of Tomorrow – Further Results from PISA 2000,* OECD, Paris.

OECD (2003b), *The PISA 2003 Assessment Framework – Mathematics, Reading, Science and Problem Solving Knowledge and Skills,* OECD, Paris.

OECD (2004a), *Learning for Tomorrow's World – First Results from PISA 2003,* OECD, Paris.

OECD (2004b), *Problem Solving for Tomorrow's World – First Measures of Cross-Curricular Competencies from PISA 2003,* OECD, Paris.

OECD (2005a), *PISA 2003 Technical Report,* OECD, Paris.

OECD (2005b), *PISA 2003 Data Analysis Manual,* OECD, Paris.

OECD (2006), *Assessing Scientific, Reading and Mathematical Literacy: A Framework for PISA 2006,* OECD, Paris.

OECD (2007), *PISA 2006: Science Competencies for Tomorrow's World*, OECD, Paris.

OECD (2009), *PISA 2006 Technical Report,* OECD, Paris.

Peaker, G.F. (1975), *An Empirical Study of Education in Twenty-One Countries: A Technical report. International Studies in Evaluation VIII*, Wiley, New York and Almqvist and Wiksell, Stockholm.

Rust, K.F. and **J.N.K. Rao** (1996), "Variance Estimation for Complex Surveys Using Replication Techniques", *Statistical Methods in Medical Research,* Vol. 5, Hodder Arnold, London, pp. 283-310.

Rutter, M., *et al.* (2004), "Gender Differences in Reading Difficulties: Findings from Four Epidemiology Studies", *Journal of the American Medical Association* 291, pp. 2007-2012.

Schulz, W. (2006), *Measuring the socio-economic background of students and its effect on achievement in PISA 2000 and PISA 2003*, Paper presented at the Annual Meetings of the American Educational Research Association (AERA) in San Francisco, 7-11 April.

Wagemaker, H. (1996), *Are Girls Better Readers. Gender Differences in Reading Literacy in 32 Countries*, IEA, The Hague.

Warm, T.A. (1989), "Weighted Likelihood Estimation of Ability in Item Response Theory", *Psychometrika,* Vol. 54(3), Psychometric Society, Williamsburg, VA., pp. 427-450.

Wright, B.D. and **M.H. Stone** (1979), *Best Test Design: Rasch Measurement,* MESA Press, Chicago.

Table of contents

LIST OF BOXES

PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

LIST OF FIGURES

LIST OF TABLES

PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

PISA DATA ANALYSIS MANUAL: SAS® SECOND EDITION – ISBN 978-92-64-05624-4 – © OECD 2009

User's Guide

Preparation of data files

All data files (in text format) and the SAS® control files are available on the PISA website (*www.pisa.oecd.org*).

SAS® users

By running the SAS® control files, the PISA data files are created in the SAS® format. Before starting analysis, assigning the folder in which the data files are saved as a SAS® library.

For example, if the PISA 2000 data files are saved in the folder of "c:\pisa2000\data\", the PISA 2003 data files are in "c:\pisa2003\data\", and the PISA 2006 data files are in "c:\pisa2006\data\", the following commands need to be run to create SAS® libraries:

```
libname PISA2000 "c:\pisa2000\data\";
libname PISA2003 "c:\pisa2003\data\";
libname PISA2006 "c:\pisa2006\data\";
run;
```

SAS® syntax and macros

All syntaxes and macros in this manual can be copied from the PISA website (*www.pisa.oecd.org*). The 17 SAS® macros presented in Chapter 17 need to be saved under "c:\pisa\macro\", before staring analysis. Each chapter of the manual contains a complete set of syntaxes, which must be done sequentially, for all of them to run correctly, within the chapter.

Rounding of figures

In the tables and formulas, figures were rounded to a convenient number of decimal places, although calculations were always made with the full number of decimal places.

Country abbreviations used in this manual

AUS	Australia	FRA	France	MEX	Mexico
AUT	Austria	GBR	United Kingdom	NLD	Netherlands
BEL	Belgium	GRC	Greece	NOR	Norway
CAN	Canada	HUN	Hungary	NZL	New Zealand
CHE	Switzerland	IRL	Ireland	POL	Poland
CZE	Czech Republic	ISL	Iceland	PRT	Portugal
DEU	Germany	ITA	Italy	SVK	Slovak Republic
DNK	Denmark	JPN	Japan	SWE	Sweden
ESP	Spain	KOR	Korea	TUR	Turkey
FIN	Finland	LUX	Luxembourg	USA	United States



From:
PISA Data Analysis Manual: SAS, Second Edition

Access the complete publication at:
https://doi.org/10.1787/9789264056251-en

Please cite this chapter as:

OECD (2009), “Use of Proficiency Levels”, in *PISA Data Analysis Manual: SAS, Second Edition*, OECD Publishing, Paris.

DOI: https://doi.org/10.1787/9789264056251-10-en

OECD